Differenciálegyenletek I.

Bevezető példa, alapfogalmak

Probléma

A rádium bomlási sebessége minden időpillanatban egyenesen arányos a még el nem bomlott rádium mennyiségével.
Határozzuk meg a t időpontban még el nem bomlott rádium mennyiségét, ha a kezdeti időpontban m_0 gramm rádiumunk van!

Megoldás

Jelöljük m(t)-vel a t időpontban még el nem bomlott rádium mennyiségét. Ekkor $\mathrm{m}(t+h)$ jelenti a $t+h$ időpontban még el nem bomlott rádium mennyiségét, s így

$$-(\mathrm{m}(t+h)-\mathrm{m}(t))$$

a [t, t + h] időintervallumban elbomlott rádium mennyiségét. (A negatív előjelre azért van szükség, mert a bomlás során a rádium mennyisége csökken) Az egységnyi idő alatt elbomlott rádium mennyiségét, a

$$-\frac{\mathrm{m}(t+h)-\mathrm{m}(t)}{h} \quad (h \neq 0)$$

hányadost a bomlás átlagos sebességének nevezzük; ennek határértéke $h \to 0$ esetén a bomlás t időpontbeli sebessége.Matematikai szempontból a bomlás átlagos sebessége a -m(t) függvény különbségi hányadosa, így a t időpontban a bomlási sebesség a -m(t) függvény t pontban vett differenciálhányadosa:

$$-\mathrm{m}'(t).$$

Mivel- a kísérletek szerint - a bomlás sebessége a t időpontban arányos a meg el nem bomlott rádium mennyiségével:

$$-\mathrm{m}'(t) = k\,\mathrm{m}(t), \qquad (1)$$

ahol a k (>0) arányossági tényező a bomló anyagra jellemző állandó. Meghatározása kísérleti úton történik.
Az m(t) tehát csak olyan függvény lehet, amely kielégíti az (1) egyenletet t minden értékére. Az (1) egyenlet így is írható:

$$\frac{\frac{d}{dt}\mathrm{m}(t)}{\mathrm{m}(t)} = -k,$$

azaz

$$\frac{d}{dt}\ln(\mathrm{m}(t)) = -k.$$

Ez pedig csak úgy állhat fenn, ha

$$\ln(\mathrm{m}(t)) = -k\,t + C \qquad \text{(C állandó)},$$

ahonnét

$$\mathrm{m}(t) = \mathbf{e}^{(-k\,t + C)} = \mathbf{e}^{C}\,\mathbf{e}^{(-k\,t)}.$$

Az $\mathrm{m}(t)$ függvény teljes meghatározásához még C-t, azaz az $\mathbf{e}^{C}$ szorzó értékét kell meghatároznunk. Ehhez használjuk fel, hogy a kezdeti (t = 0) időpontban ismerjük a rádium mennyiségét:

$$m_0 = \mathrm{m}(0) = \mathbf{e}^{C}\,\mathbf{e}^{(-k\,0)} = \mathbf{e}^{C}.$$

Tehát

$$\mathbf{m}(t) = m_0\,\mathrm{e}^{(-k\,t)} \quad (2)$$

A fenti (1) egyenletben az ismeretlen függvény mellett annak deriváltja is szerepelt. Az ilyen egyenletet differenciálegyenletnek nevezzük.

Definíció

A differenciálegyenlet olyan egyenlet, amely egy vagy több függvény deriváltjait vagy differenciáljait tartalmazza. A szereplő függvények egy- vagy többváltozós függvények is lehetnek. Ha a differenciálegyenlet csak egyváltozós függvények deriváltjait (un. közönséges deriváltakat) tartalmazza, akkor *közönséges differenciálegyenlet*ről, egyébként (ha a differenciálegyenlet parciális deriváltakat is tartalmaz) *parciális differenciálegyenlet*ről beszélünk.

Példák

$$\text{(a)}\ \frac{dy}{dx} = \frac{x^2}{\mathrm{y}(x)^2 \cos(\mathrm{y}(x))};\quad \text{(b)}\ \frac{dy}{dx} + \frac{du}{dx} = u + x^2\,y;$$

$$\text{(c)}\ (y-1)\,dx + x\cos(y)\,dy = 1;\quad \text{(d)}\ \frac{\partial}{\partial t}u = \frac{\partial^2}{\partial x^2}u;$$

$$\text{(e)}\ x^2\mathrm{y}'' + \mathrm{x}\mathrm{y}' + (x^2 - n^2)\mathrm{y} = 0;\quad \text{(f)}\ \frac{\partial^2}{\partial t^2}u = \frac{\partial^2}{\partial x^2}u.$$

Az (a), (b), (c) és az (e) közönséges, a (d) és az (f) parciális differenciálegyenlet.

A differenciálegyenletek osztályozása

1.***Rendüség szerint.*** A differenciálegyenlet rendje az egyenletben szereplő legmagasabb rendű derivált rendjével egyenlő. Így pl. az (a), a (b) ill. (c) alatti differenciálegyenlet elsőrendű, az (e) alatti másodrendű.

2.***Fokszám szerint***. A közönséges differenciálegyenlet lineáris (n-ed rendű), ha

$$a_n(x)\,\frac{d^n y}{d\,x^n} + a_{n-1}(x)\,\frac{d^{(n-1)} y}{d\,x^{(n-1)}} + \ldots + a_1(x)\,\frac{dy}{dx} + a_0(x)\,y = \mathrm{f}(x)$$

alakban írható fel, ahol az $a_j(x)$, j = 0,1,...n; és f(x) adott függvények és legalább az $a_n(x)$ nem az azonosan 0 függvény.

Így például az elsőrendű lineáris differenciálegyenlet általános alakja:

$$a_1(x)\frac{dy}{dx} + a_0(x)\,y = \mathrm{f}(x),$$

egy konkrét elsőrendű lineáris egyenlet:

$$\sin(x)\frac{dy}{dx} + \cos(x)\,y = \mathbf{e}^x.$$

Ha a differenciálegyenlet nem felel meg a fenti követelményeknek, akkor nem lineárisnak mondjuk.

Példa

Az y' + xy -1 = 0 egyenlet lineáris, az y" + cos(y) = 0 nem lineáris.

3. ***Homogenitás szerint.*** A d.e. homogén, ha nem tartalmaz olyan tagot, amely állandó, vagy amelyben csak a független változó szerepel (adott függvény). Egyébként a d.e. inhomogén. Példa. Az y'- 2y = 0 d.e. homogén, az y"-2y' + 4y-2x = 0 d.e. inhomogén.

A differenciálegyenlet megoldásai

Definíció

Egy függvény egy adott intervallumon a differenciálegyenlet megoldása, ha az adott intervallumon folytonos és deriváltjaival együtt azonosan kielégíti a differenciálegyenletet.

Példa

Igazoljuk, hogy az adott függvények megoldásai a megfelelő differenciálegyenletnek:

(a) $\frac{dy}{dx} = 3\,y$, $\mathrm{y}(x) = \mathbf{e}^{(3\,x)}$; (b) $\frac{d^2 u}{d\,x^2} + 16\,u = 0$, u(x) = $\cos(4\,x)$; és (c) y"+2y'+y = 0, y(x) = $x\,\mathbf{e}^{(-x)}$.

Megoldás

(a)

```
> restart:
> de:=diff(y(x),x)=3*y(x);
```

$$de := \frac{d}{dx}\,\mathrm{y}(x) = 3\,\mathrm{y}(x)$$

```
> subs(y(x)=exp(3*x),de);
```

$$\frac{d}{dx}(\mathbf{e}^{(3\,x)}) = 3\,\mathbf{e}^{(3\,x)}$$

(b)

```
> de1:=diff(u(x),x$2)+16*u(x)=0;
```

$$de1 := \left(\frac{d^2}{dx^2}\,\mathrm{u}(x)\right) + 16\,\mathrm{u}(x) = 0$$

```
> u:=x->cos(4*x);
```

$$u := x \to \cos(4\,x)$$

```
> de1;
```

$$0 = 0$$

```
> D(u)(x);
```

$$-4\sin(4x)$$

```
> (D@@2)(u)(x);
```

$$-16\cos(4x)$$

```
> '(D@@2)(u)(x)+16*u(x)'=(D@@2)(u)(x)+16*u(x);
```

$$(D^{(2)})(u)(x)+16\,u(x)=0$$

(c) Állítsuk elő a függvényt, majd képezzük a deriváltakat, végül számítsuk ki a differenciálegyenlet baloldalát:

```
> y:=x->x*exp(-x);
```

$$y := x \to x\,\mathbf{e}^{(-x)}$$

```
> d1:=diff(y(x),x);
```

$$d1 := \mathbf{e}^{(-x)} - x\,\mathbf{e}^{(-x)}$$

```
> d2:=diff(y(x),x$2);
```

$$d2 := -2\,\mathbf{e}^{(-x)} + x\,\mathbf{e}^{(-x)}$$

```
> d2+2*d1+y(x);
```

$$0$$

A kifejezés azonosan 0, tehát a tekintett függvény valóban megoldása a d.e.-nek. Rajzoljuk is fel a megoldásfüggvényt a[-1,1] intervallumon!

```
> plot(y(x),x=-1..1,thickness=2,title=`y(x)=x*exp(-x)`);
```



Az előző példákban a megoldás a változó függvényeként, **explicit alak**ban adott. Sok esetben a differenciálegyenlet megoldásanem adható meg explicit alakban, de implicit alakú megoldás megadható.Ezt szemlélteti a következő példa.

Példa

Igazoljuk, hogy az adott implicit függvény kielégíti a következő d.e.-et

Függvény: $2x^2 + y^2 - 2xy + 5x = 0$

Differenciálegyenlet: $\frac{dy}{dx} = \frac{2y - 4x - 5}{2y - 2x}$

Megoldás

Deriváljuk az implicit alakban adott függvényt:

$$4x + \frac{2y\,dy}{dx} - \frac{2x\,dy}{dx} - 2y + 5 = 0$$

$$\frac{dy\,(2y - 2x)}{dx} = 2y - 4x - 5$$

$$\frac{dy}{dx} = \frac{2\,y - 4\,x - 5}{2\,y - 2\,x}$$

Oldjuk meg a feladatot a Maple segítségével is!

```
> x:='x':y:='y':
> `első lépés`:=D(2*x^2+y^2-2*x*y+5*x=0);
```

$$\textit{első lépés} := 4\,\mathrm{D}(x)\,x + 2\,\mathrm{D}(y)\,y - 2\,\mathrm{D}(x)\,y - 2\,x\,\mathrm{D}(y) + 5\,\mathrm{D}(x) = 0$$

```
> `második lépés`:=subs(D(x)=1,`első lépés`);
```

$$\textit{második lépés} := 4\,x + 2\,\mathrm{D}(y)\,y - 2\,y - 2\,x\,\mathrm{D}(y) + 5 = 0$$

```
> `dy/dx`=solve(`második lépés`,D(y));
```

$$dy/dx = \frac{-4\,x + 2\,y - 5}{2\,(y - x)}$$

Ábrázoljuk a megoldásnak bizonyult függvényt! Az ábrát az implicitplot utasítással végeztethetjük el, amely a plots csomagban található:

```
> with(plots):
Warning, the name changecoords has been redefined

> implicitplot(2*x^2+y^2-2*x*y+5*x=0,x=-7..2,y=-7..2,color=gree
  n,thickness=2);
```



Példa

Mutassuk meg, hogy az adott, állandót tartalmazó, megoldás az állandó minden értékére kielégíti a differenciálegyenletet!

Megoldás

$y = C \sin(x)$,

A differenciálegyenlet: $\left(\frac{d^2}{dx^2} y(x)\right) + y(x) = 0$

```
> y:=x->C*sin(x);
```

$$y := x \to C \sin(x)$$

```
> dy1:=diff(y(x),x);
```

$$dy1 := C \cos(x)$$

```
> dy2:=diff(y(x),x$2);
>
```

$$dy2 := -C\sin(x)$$

```
> 'dy2+y(x)'=dy2+y(x);
```

$$dy2 + \mathrm{y}(x) = 0$$

```
> c_ertekek:=seq(-2.5+0.5*i,i=0..10):
> rajzhoz:={seq(subs(c=i,c*sin(x)),i=c_ertekek)};
```

$$rajzhoz := \{0., -2.5\sin(x), -2.0\sin(x), -1.5\sin(x), -1.0\sin(x), -0.5\sin(x), 0.5\sin(x), 1.0\sin(x), 1.5\sin(x), 2.0\sin(x), 2.5\sin(x)\}$$

```
> plot(rajzhoz,x=0..4*Pi,scaling=constrained,thickness=2,title=
  `Megoldások`);
```




```
>
```

Megoldás-típusok, kezdeti- és peremérték problémák

A differenciálegyenletek alkalmazása során az egyenlettel együtt adott egy vagy több feltétel, amelye(k)et a megoldás(ok)nak ki kell elégíteni(ök). Tipikus esetben a feltételek száma megegyezik az egyenlet rendjével. Bevezető példánknál, a

$$-m'(t) = k\, m(t)$$

differenciálegyenletnél a megoldás

$$m(t) = e^{(-k\,t + C)}$$

a k mellett C állandót is tartalmazza.(itt a k tk. adott, ha a bomlásban részt vevő anyag milyensáégét figyelembe vesszük) Mivel ez a megoldás a tetszőleges C állandó függvénye, ezt ***általános megoldás***nak nevezzük. Mint legtöbbször máskor is, bevezető problémánknál is adott egy un. ***kezdeti feltétel:*** $m(0) = m_0$.Az általános megoldásba helyettesítve a kezdeti feltételt megkapjuk a C értékét, itt $e^C = m_0$ adódott. Az így adódott

$$m(t) = m_0\, e^{(-k\,t)}$$

megoldást a diferenciálegyenlet egy partikuláris megoldásának nevezzük.

Definíció

Az n-ed rendű közönséges differenciálegyenlet **általános megoldás**a az a függvény, amely pontosan n számú, tetszőleges, egymástól független állandót (paramétert) tartalmaz és deriváltjaival együtt azonosan kielégíti a differenciálegyenletet.

Definíció

Az n-ed rendű közönséges differenciálegyenlet **partikuláris megoldás**a az a függvény, amely legfeljebb n-1 számú, egymástól független állandót tartalmaz és deriváltjaival együtt azonosan kielégíti a differenciálegyenletet.

Példa

Tekintsük az m=1 tömegű, a légellenállásnak kitett test pillanatnyi sebességét leíró elsőrendű differenciálegyenletet

$$\frac{dv}{dt} = 32 - v,$$

ahol v(t) a test t időpontbeli sebességét adja meg. Az egyenlet általános megoldása $v(t) = 32 - c\, e^{(-t)}$. Határozzuk meg a v(0) = 0 kezdeti feltételt kielégítő megoldást!

Megoldás

A kezdeti feltételt az általános megoldásba helyettesítve kapjuk: v(0) = 32 + c = 0.
Innen, c = -32, s így a kezdeti feltételt kielégítő partikuláris megoldás:

$$v(t) = 32 - 32\, e^{(-t)}$$

Mivel az elsőrendű differenciálegyenletek egy kiegészítő feltételt tartalmaznak, s ez többnyire kezdeti feltétel, a következő példa kapcsán egy magasabbrendű differenciálegyenlettel szemléltetjük a **kezdeti**- és a **peremérték** probléma közötti különbséget.

Példa

Tekintsük az y" + x = 0 másodrendű differenciálegyenletet, amely leírja a rugó végéhez rögzített, egységnyi tömeg mozgását, ahol k = 1 a rugóállandó értéke és x(t) a tömegpontnak az egyensúlyi helyzettől, azaz az x = 0 helyzettől, való távolságát jelenti a t időpontban.
A differenciálegyenlet általános megoldása

$$x(t) = A\cos(t) + B\sin(t).$$

Mivel az egyenlet másodrendű, az ismeretlen állandók meghatározásához két kiegészítő feltétel

szükséges.
(a)Tegyük fel, hogy a kiinduló helyzetben a részecske pozíciója **x(0) = 0** és a kezdeti időpontban a sebessége: **x'(0) = 1.** Ekkor ez egy **kezdeti érték probléma,** mert két kiegészítő feltételünk van ugyanarra a t időpontra, nevezetesen a t = 0-ra. A kezdeti feltételeket felhasználva, határozzuk meg a megoldást adó függvényt!
(b)Másrészt tegyük fel, hogyismerjük a részecske helyzetét két különböző időpontban: x(0) = 0 és $x\left(\frac{\pi}{2}\right) = -4$. Mivel a feltételek két különböző időpontra vonatkoznak, ezuttal **peremérték problémá**ról van szó. A peremérték feltételeket felhasználva határozzuk meg a megfelelő függvényt!

Megoldás

.(a) Szükségünk van az általános megoldás első deriváltjára:

$$x'(t) = -A\sin(t) + B\cos(t).$$

A helyettesítés révén adódik: x(0) = A = 0 és x'(0) = B = 1. Innen a megoldás: **x(t)= sin(t)**.

(b) Az általános megoldásba való helyettesítések révén adódik: x(0) = A=0 és $x(\frac{\pi}{2})$ = B = -4.

Így a peremérték-probléma megoldása: $\mathbf{x(t) = -4\sin(t)}$.

Iránymezők

A $\frac{dy}{dx} = f(x,y)$ {**y'=f(x,y)**} alakú elsőrendű differenciálegyenletek megoldásainak geometriai interpretációja fontos az ilyen egyenletekkel leírható problémák megértéséhez.
Legyen az egyenlet egy megoldása az $y = \phi(x)$ függvény. Geometriailag a megoldást a függvény grafikonja reprezentálja. Ilyenformán, ha (x,y) a görbe egy pontja, akkor az ezen pontbeli érintő meredeksége $(\frac{dy}{dx})$ adott az f(x,y) által.Azt mondhatjuk tehát, hogy a $\frac{dy}{dx} = f(x,y)$ differenciálegyenlet az f(x,y) függvény értelmezési tartományának minden pontjához egy irányt rendel. Az irányok összessége un. **iránymezőt** alkot. A keresett integrálgörbék ehhez az iránymezőhöz illeszkednek úgy, hogy az integrálgörbe $P_0(x_0, y_0)$ ponthoz tartozó érintője párhuzamos a differenciálegyenlet által a $P_0(x_0, y_0)$ -hoz rendelt irányhoz.
Az iránymezőt vonaldarabkákkal, irányvektorok rendszerével szemléltetjük.

Példa

(a) Szemléltessük a $\frac{dy}{dx} = e^{(-x)} - 2y$ differenciálegyenlet különböző megoldásait.
(b) Rajzoljuk fel az egyenlethez tartozó iránymezőt.

Megoldás

(a) Egyelőre még nem ismerjük a tekintett differenciálegyenlet megoldási módját, ezért a megoldást a Maple-re bízzuk.Definiáljuk az egyenletet, majd a **dsolve** utasítást használjuk az általános megoldás megkeresésére.

```
> y:='y':
> Diffegy:=diff(y(x),x)=exp(-x)-2*y(x);
```

$$Diffegy := \frac{d}{dx}\,y(x) = e^{(-x)} - 2\,y(x)$$

```
> Alt_megold:=dsolve(Diffegy,y(x));
```

$$Alt\_megold := y(x) = (e^{x} + \_C1)\,e^{(-2\,x)}$$

Az **assign** értékül adja a változónak a dsolve által megtalált megoldást.

```
> assign(Alt_megold):
```

A rajz utasítás képezi az ábrázolandó függvények sorozatát:

```
> rajz:={seq(subs(_C1=i,y(x)),i=-3..3)}:
> plot(rajz,x=-1/2..2,-1..1);
```



(b) Az egyenlethez rendelt iránymező megrajzolásához a **DEplot** utasítást használjuk, amely a DEtools csomagban található.

```
> with(DEtools):
```

```
> y:='y':
> DEplot(Diffegy,y(x),x=-1/2..1,y=-1..1,title=`Iránymező `);
```



A DEplot segítségével a megoldásokat azok explicit generálása nélkül is ábrázolhatjuk. Ez különösen hasznos olyankor, ha a megoldás explicit alakja nem érdekel bennünket,vagy nem is lehet azt előállítani. A következőkben éppen ezt mutatjuk be, tehát néhány kezdeti feltételnek megfelelő partikuláris megoldást az iránymezővel együtt szemléltetünk:

```
> DEplot(Diffegy,y(x),x=-1/2..1,{[0,.75],[0,.5],[0,0],[0,-.5],[
  0,-.75]},y=-1..1,title=`Iránymező partikuláris
  megoldásokkal`);
```

Iránymező partikuláris megoldásokkal



Az **arrows=none** opció használata esetén a megoldások az iránymező nélkül jelennek meg.

```
> DEplot(Diffegy,y(x),x=-1/2..1,{[0,.75],[0,.5],[0,0],[0,-.5],[
  0,-.75]},y=-1..1,arrows=none);
```



>

Elsőrendű, közönséges differenciálegyenletek

Szétválasztható változójú differenciálegyenletek

Definíció

Ha a differenciálegyenlet felírható

$$y' = f(x)\,g(y)$$

alakban , akkor **szétválasztható változójú differenciálegyenlet**nek nevezzük.

Megjegyzés: A Maple szintaktikát követve a fenti differenciálegyenletet

$$\frac{d}{dx}y(x) = f(x)\,g(y)$$

alakban írhatjuk föl.

Tétel

Ha az $f(x)$ az (a, b) intervallumon, a $g(y)$ a (c, d) intervallumon folytonos és $g(y) \neq 0$, ha $c < y < d$, akkor az

$$y' = f(x)\, g(y)$$

differenciálegyenletnek az $a < x < b$, $c < y < d$ téglalap minden egyes pontján egy és csak egy integrálgörbéje halad át.

Bizonyítás

Tegyük föl, hogy a tétel feltételei teljesülnek és létezik megoldás! Ez utóbbi arra utal, hogy a megfelelő határozatlan integrálok léteznek.

$$\frac{d}{dx}\, y(x) = f(x)\, g(y), \text{ ahol } g(y) \neq 0$$

Szétválasztva a változókat és az x-től való függést a g-nél is jelölve:

$$\frac{\frac{d}{dx}\, y(x)}{g(y(x))} = f(x)$$

Mindkét oldalt az x szerint integráljuk:

$$\int \frac{\frac{d}{dx}\, y(x)}{g(y(x))}\, dx = \int f(x)\, dx$$

Alkalmazzuk az $y = y(x)$ helyettesítést! Ekkor $\frac{d}{dx}\, y(x) = \frac{dy}{dx}$, tehát $\left(\frac{d}{dx}\, y(x)\right) dx = dy$ és így

$$\int \frac{1}{g(y)}\, dy = \int f(x)\, dx$$

Jelöljük a baloldalt $G(y)$-nal, a jobboldalt $F(x)$-szel. Ekkor

$$G(y) = F(x)$$

Megmutatjuk, hogy az y egyértelműen van meghatározva. A primitív függvény fogalma szerint

$$\frac{d}{dy}\, G(y) = \frac{1}{g(y)},$$

Tudjuk, hogy a $g(y)$-nak a fent definiált téglalapon nincs zérushelye és azt is, hogy a g folytonos. Ebből következik, hogy állandó előjelű. Ezért a $G(y)$ szigorúan monoton , tehát y egyértelműen adott.

A szétválasztható változójú differenciálegyenletet a tétel bizonyítása során követettek szerint oldjuk meg. Tehát úgy rendezzük az egyenletet, hogy az egyik oldal közvetlenül csak az y-tól, a másik pedig az x-től függjön. Ezután mindkét oldalon kijelöljük az x szerinti integrálást. Elvégezve az $y = y(x)$ helyettesítést, a változók szétválasztása teljessé válik. Ezután mindkét oldalon a megfelelő változó szerint integrálunk.

Példa

Mutassuk meg, hogy a

$$\frac{dy}{dx} = \frac{2\,y^{\left(\frac{1}{2}\right)} - 2\,y}{x}$$

differenciálegyenlet szétválasztható változójú, s oldjuk meg az egyenletet!

Megoldás

Az egyenlet szétválasztható változójú, mert

$$\frac{dy}{2\,y^{\left(\frac{1}{2}\right)} - 2\,y} = \frac{dx}{x}$$

alakban írható fel.Integráljuk az egyenlet mindkét oldalát:

$$\int \frac{1}{y^{\left(\frac{1}{2}\right)} - 2\,y}\,dy = \int \frac{1}{x}\,dx + C_1,$$

azaz:

$$\int \frac{1}{2\,y^{\left(\frac{1}{2}\right)}\left(1 - y^{\left(\frac{1}{2}\right)}\right)}\,dy = \int \frac{1}{x}\,dx + C_1.$$

A baloldali integrálban helyettesítsünk $u = 1 - y^{\left(\frac{1}{2}\right)}$-t, s akkor $du = -\frac{dy}{2\,y^{\left(\frac{1}{2}\right)}}$.

Ekkor adódik:

$$\int -\frac{1}{u}\,du = \int \frac{1}{x}\,dx + C_1,$$

s innen:

$$-\ln(u) = \ln(x) + C_1.$$

Felhasználva, hogy $-\ln(u) = \ln\left(\frac{1}{u}\right)$:

$$\frac{1}{u} = C\,x, \quad \text{ahol } C = \mathbf{e}^{c_1}$$

Visszahelyettesítés után adódik:

$$\frac{1}{1-y^{\left(\frac{1}{2}\right)}} = C\,x$$

a $\dfrac{dy}{dx} = \dfrac{2\,y^{\left(\frac{1}{2}\right)} - 2\,y}{x}$ általános megoldása.

Példa

Oldjuk meg az $\mathbf{y' = y\,(4-3\,y)-1}$ differenciálegyenletet
Megoldás. A differenciálegyenlet szétválasztható változójú, hiszen

$$\frac{dy}{y\,(4-3\,y)-1} = dx,$$

vagy másképp

$$\frac{dy}{-3\,y^2+4\,y-1} = dx$$

alakban írható fel.
Mindkét oldalon integrálunk:

$$\int \frac{1}{-3\,y^2+4\,y-1}\,dx = \int 1\,dx$$

A nevező valós gyöktényezők szorzatára bontható:

```
> factor(-3*y^2+4*y-1);
```

$$-(3\,y-1)\,(y-1)$$

A baloldali tört integrálását a bízzuk a Maple-re:

```
> Int(1/(%),y)=int(1/(%),y);
```

$$\int -\frac{1}{(3\,y-1)\,(y-1)}\,dy = \frac{1}{2}\ln(3\,y-1) - \frac{1}{2}\ln(y-1)$$

A jobboldali integrál: x+c

```
> rhs(%)=x+c;
```

$$\frac{1}{2}\ln(3\,y-1) - \frac{1}{2}\ln(y-1) = x+c$$

Az isolate eljárás segítségével y-t kifejezhetjük:

```
> isolate(%,y);
```

$$y = \frac{-1+\mathbf{e}^{(2\,x+2\,c)}}{\mathbf{e}^{(2\,x+2\,c)}-3}$$

Oldjuk meg az egyenletet a Maple segítségével is!

```
> de:=diff(y(x),x)=y(x)*(4-3*y(x))-1;
```

$$de := \frac{d}{dx}\,\mathrm{y}(x) = \mathrm{y}(x)\,(4-3\,\mathrm{y}(x))-1$$

```
> dsolve(de,y(x));
```

$$y(x) = \frac{-1 + e^{(2x)}\_C1}{e^{(2x)}\_C1 - 3}$$

Példa

Oldjuk meg az $y' = -x\,y^2$ differenciálegyenletet!

```
> y:='y':
> de:=diff(y(x),x)=-x*y(x)^2;
```

$$de := \frac{d}{dx}\,y(x) = -x\,y(x)^2$$

```
> altmegold:=dsolve(de,y(x));
```

$$altmegold := y(x) = \frac{2}{x^2 + 2\_C1}$$

```
> DEplot(de,y(x),x=-2..2,{[0,1],[0,2],[0,3]},y=0..3.2);
```




```
> y(x):='y(x)':y:='y':
> with(DEtools):
```

Példa

Oldjuk meg az $\mathbf{y' = 2\sqrt{y-1}}$ differenciálegyenletet!

```
> assume(y>1):
> de:=diff(y(x),x)=2*sqrt(y(x)-1);
```

$$de := \frac{d}{dx}\,\mathrm{y\sim}(x) = 2\sqrt{\mathrm{y\sim}(x)-1}$$

```
> altmegold:=dsolve(de,y(x));
```

$$altmegold := x - \sqrt{\mathrm{y\sim}(x)-1} + \_C1 = 0$$

```
> isolate(altmegold,y(x));
```

$$\mathrm{y\sim}(x) = (x + \_C1)^2 + 1$$

```
> DEplot(de,y(x),x=1.01..5,{[2,2],[3,4],[4,5.6],[5,10]},y=1.
05..10);
```




```
>
```

Példa

Newton hülési törvénye szerint a lehűlő test T(t) hőmérsékletének változása egyenesen arányos a lehűlő test és környezete hőmérsékletének különbségével, azaz

$$\frac{dT}{dt} = k\,(T - Tk),\quad \mathrm{T}(0) = T_0,$$

ahol T_0 a test hőmérséklete a $t = 0$ kezdeti időpontban, T_k pedig a környezet hőmérséklete. A süteményt a 250°C hőmérsékletű sütőből , lévén színe szép barnás, jól megsült, kivesszük. A konyha hőmérséklete 20°C. 15 perc múlva a sütemény hőmérséklete 70°C. Mennyi idő múlva ehetünk belőle, ha 30°C-on ehető?

Megoldás

1. "Kézi" megoldás:

A $\frac{dT}{dt} = k\,(T - Tk)$ egyenlet un. szétválasztható változójú, azaz

$$\frac{dT}{T - Tk} = k\,dt \qquad (1)$$

alakúra hozható. Mindkét oldalt a megfelelő változó szerint integrálva:

$$\int \frac{1}{T - Tk}\,dT = \int k\,dt.$$

Innen:

$$\ln\left(\frac{|T - Tk|}{C}\right) = k\,t$$

Ebből:

$$|T - Tk| = C\,\mathbf{e}^{(k\,t)},\qquad 0 < C,\ \text{állandó.}$$

Ezért:

$$T = C\,\mathbf{e}^{(k\,t)} + Tk$$

A differenciálegyenlet általános megoldása.

Mivel $\mathrm{T}(0) = T_0$, tehát $C\,\mathbf{e}^0 + Tk = T_0$, $C = T_0 - Tk$,

s így

$$\mathrm{T}(t) = (T_0 - Tk)\,\mathbf{e}^{(k\,t)} + Tk,$$

a hűlési folyamatot leíró függvény.

2. Nézzük mindezt a Maple segítségével:

```
> restart:
> egy:= int(1/(T-Tk),T)-ln(C)=int(k,t);
```

$$egy := \ln(T - Tk) - \ln(C) = k\,t$$

```
> isolate(egy,T);
```

$$T = \mathbf{e}^{(k\,t)}\,C + Tk$$

```
> T:=unapply(rhs(%),t);
```

$$T := t \rightarrow \mathbf{e}^{(k\,t)}\,C + Tk$$

```
> C:=solve(T(0)=T0,C);
```

$$C := -Tk + T0$$

```
> 'T(t)'=T(t);
```

$$T(t) = e^{(k\,t)}\,(-Tk + T0) + Tk$$

3. Most oldjuk meg a problémát a Maple beépített, **dsolve**, eljárásával. Egyuttal adjuk meg a kérdésre is a választ:

```
> T:='T':
```

```
> de:=diff(T(t),t)=k*(T(t)-Tk);
```

$$de := \frac{d}{dt}\,T(t) = k\,(T(t) - Tk)$$

```
> altmegold:=dsolve({de,T(0)=T0},T(t));
```

$$altmegold := T(t) = e^{(k\,t)}\,(-Tk + T0) + Tk$$

```
> lepes_1:=subs({T0=250,Tk=20},rhs(altmegold));
```

$$lepes\_1 := 230\,e^{(k\,t)} + 20$$

```
> k:=solve(subs(t=15,lepes_1)=70);
```

$$k := \frac{1}{15}\ln\left(\frac{5}{23}\right)$$

```
> simplify(lepes_1,exp);
```

$$230\left(\frac{5}{23}\right)^{\left(\frac{t}{15}\right)} + 20$$

```
> T:=unapply(%,t);
```

$$T := t \to 230\left(\frac{5}{23}\right)^{(1/15\,t)} + 20$$

```
> plot(T(t),t=20..35);
```




```
> jo:=solve(T(t)=30);
```

$$jo := -\frac{15\ln(23)}{\ln\left(\frac{5}{23}\right)}$$

```
> evalf(jo);
```

$$30.81957927$$

Tehát kb.fél óra múlva ehetünk a sütiből.

```
>
```

Homogén fokszámú differenciálegyenletek

Definíció

.Az f(x,y) kétváltozós függvényt n-ed fokú homogén függvénynek nevezzük, ha

$$\mathbf{f(\lambda x,\lambda y) = \lambda^n f(x, y)}$$

Például az

$$f(x, y) = x^4 - x^3 y$$

függvény negyedfokú homogén függvény, mert

$$f(\lambda x, \lambda y) = (\lambda x)^4 - (\lambda x)^3 \lambda y = \lambda^4 f(x, y).$$

Definíció

Az **$M(x, y)\, dx + N(x, y)\, dy = 0$** elsőrendű differenciálegyenletet akkor nevezzük homogén fokszámúnak, ha az M(x,y) és N(x,y) ugyanolyan fokszámú homogén függvények.

A homogén fokszámú differenciálegyenlet az $\frac{y}{x} = t$, azaz $y = x\, t$ helyettesítéssel visszavezethető szétválasztható változójú differenciálegyenletre. Ha az $y = x\, t$ helyettesítést alkalmazzuk, akkor $y' = \frac{dy}{dx} = t + \frac{x\, dt}{dx}$,

ill.

$$dy = t\, dx + x\, dt.$$

Ha a differenciálegyenlet homogén fokszámú, akkor kellő rendezés után

$$y' = g\left(\frac{y}{x}\right)$$

alakra hozható.

Példa. Oldjuk meg az **$(x^2 - y^2)\, dx + x\, y\, dy = 0$** differenciálegyenletet!

Megoldás. Megmutatjuk, hogy az egyenlet homogén fokszámú:

```
> egy:=(x^2-y(x)^2)*dx+x*y(x)*dy=0;
```

$$egy := (x^2 - y(x)^2)\, dx + x\, y(x)\, dy = 0$$

```
> ### WARNING: persistent store makes one-argument readlib obsolete
  readlib(isolate):
```

Rendezzük az egyenletet:

```
> isolate(egy,dy);
```

$$dy = -\frac{(x^2 - y(x)^2)\, dx}{x\, y(x)}$$

```
> %/dx;
```

$$\frac{dy}{dx} = -\frac{x^2 - y(x)^2}{x\, y(x)}$$

```
> lhs(%)=expand(rhs(%));
```

$$\frac{dy}{dx} = -\frac{x}{y(x)} + \frac{y(x)}{x}$$

Az egyenlet valóban homogén fokszámú!

Végezzük el a helyettesítést:

$$\frac{y}{x} = t; \;\; y = t(x)\, x = t\, x;$$

$$y' = t + \frac{dt}{dx}\, x = -\frac{1}{t} + t$$

Innen

$$t\, dt = -\frac{dx}{x} \text{ ,azaz}$$

$$\int t\,dt = \int -\frac{1}{x}\,dx$$

$$\frac{t^2}{2} = -\ln(|x|) + \ln(|c|) = \ln\left(\left|\frac{c}{x}\right|\right)$$

Ebből:

$$e^{\left(\frac{t^2}{2}\right)} = \frac{c}{x}$$

Visszahelyettesítve:

$$e^{\left(\frac{\left(\frac{y}{x}\right)^2}{2}\right)} = \frac{c}{x}$$

Az explicit alak: $\mathbf{y = x\sqrt{2\ln\left(\frac{c}{x}\right)}}$ ill. $\mathbf{y = - x\sqrt{2\ln\left(\frac{c}{x}\right)}}$

Oldjuk meg az egyenletet Maple segítségével is:

```
> de:=diff(y(x),x)=y(x)/x-x/y(x);
>
```

$$de := \frac{d}{dx}\,y(x) = -\frac{x}{y(x)} + \frac{y(x)}{x}$$

```
> dsolve(de,y(x));
```

$$y(x) = \sqrt{-2\ln(x) + \_C1}\;x,\; y(x) = -\sqrt{-2\ln(x) + \_C1}\;x$$

```
>
```

Lineáris differenciálegyenletek

DEFINÍCIÓ. Az $\mathbf{y' = P(x)\,y + Q(x)}$ alakban felírható differenciálegyenletet lineáris differenciálegyenletnek nevezzük.

A $P(x)$ és a $Q(x)$ adott függvények. Ha Q(x) az azonosan 0 függvény, akkor a differenciálegyenlet homogén, egyébként inhomogén.

Előbb a homogén lineáris differenciálegyenlet megoldásával foglalkozunk, majd ennek felhasználásával oldjuk meg az inhomogén lineáris differenciálegyenleteket.

1. Tekintsük tehát az

$$y' = P(x)\,y \text{ azaz } \frac{dy}{dx} = P(x)\,y$$

homogén egyenletet. Ez szétválasztható változójú, így

$$\int \frac{1}{y}\,dy = \int P(x)\,dx, \text{ azaz}$$

$$\ln(|y|) - \ln(|c|) = \int P(x)\,dx$$

vagyis

$$\ln\left(\left|\frac{y}{c}\right|\right) = \int P(x)\,dx.$$

Innen $$\frac{y}{c} = \mathbf{e}^{\left(\int P(x)\,dx\right)},$$

azaz $$y = c\,\mathbf{e}^{\left(\int P(x)\,dx\right)},$$

ahol c tetszőleges 0-tól különböző valós szám. Ez tehát a homogén egyenlet általános megoldása.

2. Most tekintsük az

$$y' = P(x)\,y + Q(x) \qquad (1)$$

inhomogén lineáris differenciálegyenletet. Ha ebben Q(x) helyére 0-t írunk, akkor az inhomogén differenciálegyenlethez tartozó homogén lineáris differenciálegyenletet kapjuk:

$$Y' = P(x)\,Y$$

(ennek ismeretlenjét azért jelöltük Y-nal, mert y-t fenntartjuk az inhomogén differenciálegyenlet megoldásának a jelölésére) Az előbbiek szerint ezen homogén lineáris differenciálegyenlet általános megoldása:

$$y = c\,\mathbf{e}^{\left(\int P(x)\,dx\right)}.$$

Keressük az (1) inhomogén egyenlet megoldását is ilyen alakban, de abban a c helyére x-nek olyan c(x) függvényét tegyük, hogy

$$y = c(x)\,\mathbf{e}^{\left(\int P(x)\,dx\right)} \qquad (2)$$

már az inhomogén egyenlet megoldása legyen. Feladatunk így a c(x) függvény meghatározása. Ehhez helyettesítsük a (2) függvényt az (1) egyenletbe. Ekkor kapjuk, hogy

$$c'(x)\mathbf{e}^{\left(\int P(x)\,dx\right)} + c(x)\,\mathbf{e}^{\left(\int P(x)\,dx\right)} P(x) = P(x)\,c(x)\,\mathbf{e}^{\left(\int P(x)\,dx\right)} + Q(x).$$

Ebből

$$c'(x) = Q(x)\,\mathbf{e}^{\left(-\int P(x)\,dx\right)}.$$

Innen

$$c(x) = \int Q(x)\,\mathbf{e}^{\left(-\int P(x)\,dx\right)}\,dx + c.$$

Így az (1) inhomogén differenciálegyenlet általános megoldása

$$y = \left(\int Q(x)\,\mathbf{e}^{\left(-\int P(x)\,dx\right)}\,dx + c\right)\mathbf{e}^{\left(\int P(x)\,dx\right)}$$

Ezt a módszert, amellyel a homogén differenciálegyenlet megoldásából az inhomogén lineáris differenciálegyenlet megoldását megkerestük az állandó variálásának nevezzük. A módszer L.Euler-től származik.

PÉLDA. Oldjuk meg az

$$y' = \frac{2}{x}\,y + x^2 + 1$$

differenciálegyenletet. Adjuk meg az $y(1) = 1$ kezdeti feltételt kielégítő partikuláris megoldást is!

MEGOLDÁS. A megoldandó egyenlet lineáris inhomogén.Először megoldjuk a hozzá tartozó

homogén egyenletet, majd az Euler-féle állandó variálásának módszerével megadjuk az eredeti egyenlet általános megoldását.
A homogén egyenlet:

$$y' = \frac{2}{x}\,y, \text{ azaz}$$

$$\frac{dy}{y} = \frac{2\,dx}{x}.$$

Mindkét oldalt integrálva:

$$\int \frac{1}{y}\,dy = \int \frac{2}{x}\,dx.$$

Ebből:

$$\ln(|y|) = \ln(|c\,x^2|).$$

Az y-t kifejezve:

$$\boldsymbol{y = c\,x^2}.$$

Megkaptuk a homogén egyenlet általános megoldását. Most az *állandó variálása* következik.

$$y = c(x)\,x^2$$

$$y' = c'(x)\,x^2 + 2\,x\,c(x)$$

Ezeket az eredeti egyenletbe helyettesítve:

$$c'(x)x^2 + 2\,x\,c(x) = 2\,x\,c(x) + x^2 + 1.$$

Ebből

$$c'(x) = 1 + \frac{1}{x^2},$$

tehát integrálva

$$c(x) = x - \frac{1}{x} + c.$$

Így az inhomogén egyenlet általános megoldása:

$$\boldsymbol{y = \left(x - \frac{1}{x} + c\right)x^2 = x^3 - x + c\,x^2}.$$

A keresett partikuláris megoldást úgy kapjuk meg, hogy az általános megoldásba helyettesítjük a kezdeti feltételt:

$$y(1) = 1^3 - 1 + c\,1^2 = 1.$$

Ebből $c = 1$,

tehát a keresett partikuláris megoldás: $\boldsymbol{y = x^3 - x + x^2}$

Nézzük meg most mindezt Maple segítségével:

```
> restart:
> de:=diff(y(x),x)=2/x*y(x)+x^2+1;
```

$$de := \frac{d}{dx}\,y(x) = \frac{2\,y(x)}{x} + x^2 + 1$$

```
> Alt_megold:=dsolve(de,y(x));
```

$$Alt\_megold := y(x) = \left(x - \frac{1}{x} + \_C1\right)x^2$$

```
> assign(Alt_megold);
> rajz:={seq(subs(_C1=i,y(x)),i=-3..3)}:
> plot(rajz,x=-1/2..2,-1..1,thickness=3,title=`Az általános
  megoldás néhány görbéje`);
```



Szemléltessük az egyenlethez tartozó iránymezőt:

```
> with(DEtools):
> y:='y':
> DEplot(de,y(x),x=-1/2..1.6,y=-1..1,title=`Iránymező `);
```




```
> restart:
> de:=diff(y(x),x)=2/x*y(x)+x^2+1;
```

$$de := \frac{d}{dx}\,\mathrm{y}(x) = \frac{2\,\mathrm{y}(x)}{x} + x^2 + 1$$

```
> with(DEtools):
> `partmegoldás`:=dsolve({de,y(1)=1},y(x),explicit);
```

$$partmegoldás := \mathrm{y}(x) = \left(x - \frac{1}{x} + 1\right)x^2$$

```
> assign(`partmegoldás`):
> plot(y(x),x=-2..1.2,color=blue,thickness=2,title=`A
  partikuláris megoldás görbéje a jellemző intervallumon`);
```



>

A Bernoulli-féle differenciálegyenlet

A Bernoulli-féle differenciálegyenlet nem lineáris, általános alakja:

$$y' + p(x)\,y = q(x)\,y^n.$$

Ezt a differencálegyenletet Jacob Bernoulli (1654-1705) állította fel 1695-ben, megoldását Johann Bernoulli (1667-1748) adta meg 1697-ben. A Bernoulli-féle differenciálegyenlet helyettesítéssel lineárissá tehető.Ha n = 0, vagy n=1, akkor az egyenlet lineáris. Ha $n \neq 0$, vagy
1, akkor az egyenlet helyettesítéssel lineárissá tehető. Legyen

$$w = y^{(1-n)},$$

ekkor

$$\frac{dw}{dx} = \frac{(1-n)\,y^{(-n)}\,dy}{dx}.$$

Ezeket az egyenletbe helyettesítve:

$$\frac{y^n\,dw}{(1-n)\,dx} + p(x)\,y^n\,w = q(x)\,y^n.$$

Az egyenletet $\frac{1-n}{y^n}$ -nel szorozva elsőrendű lineáris egyenletet kapunk:

$$\frac{dw}{dx} + (1-n)\,\mathrm{p}(x)\,w = (1-n)\,\mathrm{q}(x),$$

amelyet w(x)-re a lineáris egyenleteknél látott módon megoldunk, majd az y(x)-et az $w = y^{(1-n)}$ vagy $y = w^{\left(\frac{1}{1-n}\right)}$ összefüggés felhasználásával kapjuk meg.

PÉLDA. Oldjuk meg a $\frac{dy}{dx} + \frac{y}{x} = \frac{1}{x\,y^2}$, $0 < x$ Bernoulli-féle egyenletet!

MEGOLDÁS. Ebben az esetben $\mathrm{p}(x) = \frac{1}{x}$, $\mathrm{q}(x) = \frac{1}{x}$ és $n = -2$. Ezért a helyettesítés

$$w = y^{(1-(-2))} = y^3,$$

ami azt jelenti, hogy

$$\frac{dw}{dx} = \frac{3\,y^2\,dy}{dx}.$$

A deriváltat az egyenletbe helyettesítve:

$$\frac{1}{3\,y^2}\,\frac{dw}{dx} + \frac{y}{x} = \frac{1}{x\,y^2}.$$

Szorozzunk $3\,y^2$-tel:

$$\frac{dw}{dx} + \frac{3\,y^3}{x} = \frac{3}{x}.$$

Természetesen az $w = y^3$ helyettesítést is el kell végeznünk, s ekkor w-ra elsőrendű, lineáris egyenletet kapunk:

$$\frac{dw}{dx} + \frac{3\,w}{x} = \frac{3}{x}.$$

Oldjuk ezt meg !

```
> w:='w':y:='y':
> de:=diff(w(x),x)+ 3*w(x)/x=3/x;
```

$$de := \left(\frac{d}{dx}\,\mathrm{w}(x)\right) + \frac{3\,\mathrm{w}(x)}{x} = \frac{3}{x}$$

```
> alt_megold:=dsolve(de,w(x));
```

$$alt\_megold := \mathrm{w}(x) = 1 + \frac{\_C1}{x^3}$$

```
> alt_megold:=subs(w(x)=y(x)^3,alt_megold);
```

$$alt\_megold := \mathrm{y}(x)^3 = 1 + \frac{\_C1}{x^3}$$

A Maple segítségével természetesen az egyenletet közvetlenül is megoldhatjuk.

```
> de1:=diff(y(x),x)+y(x)/x = 1/(x*y(x)^2);
```

$$de1 := \left(\frac{d}{dx}\,\mathrm{y}(x)\right) + \frac{\mathrm{y}(x)}{x} = \frac{1}{x\,\mathrm{y}(x)^2}$$

```
> dsolve(de1,y(x),implicit);
```

$$y(x)^3 - 1 - \frac{\_C1}{x^3} = 0$$